NAG Library Routine Document

S30NAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

S30NAF computes the European option price given by Heston's stochastic volatility model.

2 Specification

```
    SUBROUTINE S30NAF(CALPUT, M, N, X, S, T, SIGMAV, KAPPA, CORR, VAR0, ETA,
   1                  GAMMA, R, Q, P, LDP, IFAIL)
    INTEGER           M, N, LDP, IFAIL
    double precision  X(M), S, T(N), SIGMAV, KAPPA, CORR, VAR0, ETA, GAMMA,
   1                  R, Q, P(LDP,N)
    CHARACTER*1       CALPUT
```

3 Description

S30NAF computes the price of a European option using Heston's stochastic volatility model (see Heston (1993)). The return on the asset price, S, is

$$\frac{dS}{S} = (r-q)dt + \sqrt{v_t}dW_t^{(1)}$$

and the instantaneous variance, v_t, is defined by a mean-reverting square root stochastic process,

$$dv_t = \kappa(\eta - v_t)dt + \sigma_v\sqrt{v_t}dW_t^{(2)},$$

where r is the risk-free annual interest rate; q is the annual dividend rate; σ_v is the volatility of the volatility,$\sqrt{v_t}$; κ is the mean reversion rate; η is the long term variance. $dW_t^{(i)}$, for $i = 1, 2$, denotes two correlated standard Brownian motions with

$$\mathrm{Cov}\left[dW_t^{(1)}, dW_t^{(2)}\right] = \rho dt.$$

The option price is computed by evaluating the integral transform given by Lewis (2000) using the form of the characteristic function discussed by Albrecher *et al.* (2007), see also Kilin (2006). For a given strike price, X, and time to expiry, T, the price of a European call is

$$P_{\text{call}} = Se^{-qT} - Xe^{-rT}\frac{1}{\pi}\int_{0+i/2}^{\infty+i/2} e^{ik\bar{X}}\frac{\hat{H}(k,v,T)}{k^2-ik}dk, \tag{1}$$

where $\bar{X} = \ln(S/X) + (r-q)T$ and

$$\hat{H}(k,v,t) = \exp\left(\frac{2\kappa\eta}{\sigma_v^2}\left[tg - \ln\left(\frac{1-he^{-\xi t}}{1-h}\right)\right] + v_t g\left[\frac{1-e^{-\xi t}}{1-he^{-\xi t}}\right]\right),$$
$$g = \tfrac{1}{2}(b-\xi),$$
$$h = \tfrac{b-\xi}{b+\xi},$$
$$\xi = \left[b^2 + 4\frac{(k^2-ik)}{\sigma_v^2}\right]^{\frac{1}{2}},$$
$$b = \tfrac{2}{\sigma_v^2}\left[(1-\gamma+ik)\rho\sigma_v + \sqrt{\kappa^2 - \gamma(1-\gamma)\sigma_v^2}\right]$$

with $t = \sigma_v^2 T/2$. Here γ is the risk aversion parameter of the representative agent with $0 \leq \gamma \leq 1$. The value $\gamma = 1$ corresponds to $\lambda = 0$ (the market price of risk) in Heston's paper (see Rouah and Vainberg (2007)).

The price of a put option is obtained by put-call parity.

4 References

Albrecher H, Mayer P, Schoutens W and Tistaert J (2007) The little Heston trap *Wilmott Magazine* **January 2007** 83–92

Heston S (1993) A closed-form solution for options with stochastic volatility with applications to bond and currency options **6** 347–343 Review of Financial Studies

Kilin F (2006) Accelerating the calibration of stochastic volatility models *MPRA Paper No. 2975*http://mpra.ub.uni-muenchen.de/2975/

Lewis A L (2000) Option Valuation Under Stochastic Volatility *Finance Press, USA*

Rouah F D and Vainberg G (2007) *Option Pricing Models and Volatility using Excel-VBA* John Wiley and Sons, Inc

5 Parameters

1: CALPUT – CHARACTER*1 *Input*

On entry: determines whether the option is a call or a put.

CALPUT = 'C'
A call. The holder has a right to buy.

CALPUT = 'P'
A put. The holder has a right to sell.

Constraint: CALPUT = 'C' or 'P'.

2: M – INTEGER *Input*

On entry: the number of strike prices to be used.

Constraint: $\mathrm{M} \geq 1$.

3: N – INTEGER *Input*

On entry: the number of times to expiry to be used.

Constraint: $\mathrm{N} \geq 1$.

4: X(M) – ***double precision*** array *Input*

On entry: $\mathrm{X}(i)$ must contain X_i, the ith strike price, for $i = 1, 2, \ldots, \mathrm{M}$.

Constraint: $\mathrm{X}(i) \geq z$ and $\mathrm{X}(i) \leq 1/z$, where $z = \mathrm{X02AMF}()$, the safe range parameter, for $i = 1, 2, \ldots, \mathrm{M}$.

5: S – ***double precision*** *Input*

On entry: S, the price of the underlying asset.

Constraint: $\mathrm{S} \geq z$ and $\mathrm{S} \leq 1/z$, where $z = \mathrm{X02AMF}()$, the safe range parameter.

6: T(N) – ***double precision*** array *Input*

On entry: $\mathrm{T}(i)$ must contain T_i, the ith time, in years, to expiry, for $i = 1, 2, \ldots, \mathrm{N}$.

Constraint: $\mathrm{T}(i) \geq z$, where $z = \mathrm{X02AMF}()$, the safe range parameter, for $i = 1, 2, \ldots, \mathrm{N}$.

7: SIGMAV – ***double precision*** *Input*

On entry: the volatility, σ_v, of the volatility process, $\sqrt{v_t}$. Note that a rate of 20% should be entered as 0.2.

Constraint: SIGMAV > 0.0.

8: KAPPA – ***double precision*** *Input*

On entry: κ, the long term mean reversion rate of the volatility.

Constraint: KAPPA > 0.0.

9: CORR – ***double precision*** *Input*

On entry: the correlation between the two standard Brownian motions for the asset price and the volatility.

Constraint: $-1.0 \leq$ CORR ≤ 1.0.

10: VAR0 – ***double precision*** *Input*

On entry: the initial value of the variance, v_t, of the asset price.

Constraint: VAR0 ≥ 0.0.

11: ETA – ***double precision*** *Input*

On entry: η, the long term mean of the variance of the asset price.

Constraint: ETA > 0.0.

12: GAMMA – ***double precision*** *Input*

On entry: the risk aversion parameter, γ, of the representative agent.

Constraint: $0.0 \leq$ GAMMA ≤ 1.0.

13: R – ***double precision*** *Input*

On entry: r, the annual risk-free interest rate, continuously compounded. Note that a rate of 5% should be entered as 0.05.

Constraint: R ≥ 0.0.

14: Q – ***double precision*** *Input*

On entry: q, the annual continuous yield rate. Note that a rate of 8% should be entered as 0.08.

Constraint: Q ≥ 0.0.

15: P(LDP,N) – ***double precision*** array *Output*

On exit: the $m \times n$ array P contains the computed option prices.

16: LDP – INTEGER *Input*

On entry: the first dimension of the array P as declared in the (sub)program from which S30NAF is called.

Constraint: LDP $\geq$ M.

17: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, CALPUT $\neq$ 'C' or 'P'.

IFAIL $= 2$

On entry, M ≤ 0.

IFAIL $= 3$

On entry, N ≤ 0.

IFAIL $= 4$

On entry, X$(i) < z$ or X$(i) > 1/z$, where $z =$ X02AMF(), the safe range parameter.

IFAIL $= 5$

On entry, S $< z$ or S $> 1/z$, where $z =$ X02AMF(), the safe range parameter.

IFAIL $= 6$

On entry, T$(i) < z$, where $z =$ X02AMF(), the safe range parameter.

IFAIL $= 7$

On entry, SIGMAV ≤ 0.0.

IFAIL $= 8$

On entry, KAPPA ≤ 0.0.

IFAIL $= 9$

On entry, $|$CORR$| > 1.0$.

IFAIL $= 10$

On entry, VAR0 < 0.0.

IFAIL $= 11$

On entry, ETA ≤ 0.0.

IFAIL $= 12$

On entry, GAMMA < 0.0,
or GAMMA > 1.0.

IFAIL $= 13$

On entry, R < 0.0.

IFAIL = 14

On entry, Q < 0.0.

IFAIL = 16

On entry, LDP < M.

IFAIL = 17

Quadrature has not converged; an approximate value of the derivative has/has not been returned in P.

7 Accuracy

The accuracy of the output is determined by the accuracy of the numerical quadrature used to evaluate the integral in (1). An adaptive method is used which evaluates the integral to within a tolerance of $\max\left(10^{-8}, 10^{-10} \times |I|\right)$, where $|I|$ is the absolute value of the integral.

8 Further Comments

None.

9 Example

This example computes the price of a European call using Heston's stochastic volatility model. The time to expiry is 6 months, the stock price is 100 and the strike price is 100. The risk-free interest rate is 5% per year, the volatility of the variance, σ_v, is 22.5% per year, the mean reversion parameter, κ, is 2.0, the long term mean of the variance, η, is 0.01 and the correlation between the volatility process and the stock price process, ρ, is 0.0. The risk aversion parameter, γ, is 1.0 and the initial value of the variance, VAR0, is 0.01.

9.1 Program Text

```
*     S30NAF Example Program Text
*     Mark 22 Release. NAG Copyright 2007.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          LDP, MMAX, NMAX
      PARAMETER        (LDP=50,MMAX=50,NMAX=50)
*     .. Local Scalars ..
      DOUBLE PRECISION CORR, ETA, GAMMA, KAPPA, Q, R, S, SIGMAV, VAR0
      INTEGER          I, IFAIL, J, M, N
      CHARACTER        PUT
*     .. Local Arrays ..
      DOUBLE PRECISION P(LDP,NMAX), T(NMAX), X(MMAX)
*     .. External Subroutines ..
      EXTERNAL         S30NAF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'S30NAF Example Program Results'
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Heston''s Stochastic volatility Model'
*     Skip heading in data file
      READ (NIN,*)
*     Read problem parameters.
      READ (NIN,*) PUT
      READ (NIN,*) S, R, Q
      READ (NIN,*) KAPPA, ETA, VAR0, SIGMAV, CORR, GAMMA
*     Read M, N
      READ (NIN,*) M, N
*
      IF (M.LE.MMAX .AND. N.LE.NMAX) THEN
*        Read array of strike/exercise prices, X
```

```
         READ (NIN,*) (X(I),I=1,M)
         READ (NIN,*) (T(I),I=1,N)
*
         IFAIL = 1
*
         CALL S30NAF(PUT,M,N,X,S,T,SIGMAV,KAPPA,CORR,VAR0,ETA,GAMMA,R,Q,
     +               P,LDP,IFAIL)
*
         IF (IFAIL.EQ.0) THEN
            IF (PUT.EQ.'C' .OR. PUT.EQ.'c') THEN
               WRITE (NOUT,*) 'European Call :'
            ELSE IF (PUT.EQ.'P' .OR. PUT.EQ.'p') THEN
               WRITE (NOUT,*) 'European Put :'
            END IF
            WRITE (NOUT,'(A,1X,F8.4)') '  Spot                   = ', S
            WRITE (NOUT,'(A,1X,F8.4)') '  Volatility of vol      = ',
     +        SIGMAV
            WRITE (NOUT,'(A,1X,F8.4)') '  Mean reversion         = ',
     +        KAPPA
            WRITE (NOUT,'(A,1X,F8.4)') '  Correlation            = ',
     +        CORR
            WRITE (NOUT,'(A,1X,F8.4)') '  Variance               = ',
     +        VAR0
            WRITE (NOUT,'(A,1X,F8.4)') '  Mean of variance       = ',
     +        ETA
            WRITE (NOUT,'(A,1X,F8.4)') '  Risk aversion          = ',
     +        GAMMA
            WRITE (NOUT,'(A,1X,F8.4)') '  Rate                   = ', R
            WRITE (NOUT,'(A,1X,F8.4)') '  Dividend               = ', Q
*
            WRITE (NOUT,*)
            WRITE (NOUT,*) '   Strike    Expiry    Option Price'
*
            DO 40 I = 1, M
               DO 20 J = 1, N
                  WRITE (NOUT,99999) X(I), T(J), P(I,J)
   20          CONTINUE
   40       CONTINUE
         ELSE
            WRITE (NOUT,*)
            WRITE (NOUT,99998) IFAIL
         END IF
      END IF
*
99999 FORMAT (1X,2(F9.4,1X),6X,F9.4)
99998 FORMAT (1X,' ** S30AAF returned with IFAIL = ',I5)
      END
```

9.2 Program Data

```
S30NAF Example Program Data
 'C'                          : Call = 'C', Put = 'P'
 100.0 0.05 0.0               : S, R, Q
 2.0 0.01 0.01 0.225 0.0 1.0 : KAPPA, ETA, VAR0, SIGMAV, CORR, GAMMA
 1 1                          : M, N
 100.0                        : X(I), I = 1,2,...N
 0.5                          : T(I), I = 1,2,...M
```

9.3 Program Results

```
 S30NAF Example Program Results

 Heston's Stochastic volatility Model
 European Call :
  Spot                   =  100.0000
  Volatility of vol      =    0.2250
  Mean reversion         =    2.0000
  Correlation            =    0.0000
  Variance               =    0.0100
  Mean of variance       =    0.0100
```

```
 Risk aversion          =    1.0000
 Rate                   =    0.0500
 Dividend               =    0.0000

   Strike    Expiry    Option Price
 100.0000    0.5000          4.0851
```